

## MoveItemsBetweenListboxes

(Q) Can I move easily items from one listbox to another ?

tags:  
[HowTo](#)

(R) Let's implement the classical method of moving either the selected items or all the items from one listbox to another and reciprocally.

A first listbox is loaded from the database. The second one is empty.

The goal is to load the second listbox with a selection of items from the first one, thanks to 4 buttons:

- > and < move all selected items from one listbox to the other one and remove them from the original one.
- >> and << empty the original box and move all its items to the other one.

### Solution

It consists in putting on a form:

- 2 **listboxes** called *LeftList* and *RightList*
- 4 buttons called *MoveOneRight* (>), *MoveAllRight* (>>), *MoveOneLeft* (<), *MoveAllLeft* (<<)

The code associated with the events is mainly a matter of arrays redimensioning, trimming and housekeeping. Both listboxes must be reinitialized from the event macro after each button activation to remain synchronized.

Let's assume next table:

- Products table

Fields	Field Type	Primary
CategoryID	BigInt	
ProductName	Text	
QuantityPerUnit	Text	
ReorderLevel	Integer	
SupplierID	BigInt	
UnitPrice	Number	
UnitsInStock	Integer	
UnitsOnOrder	Integer	
Discontinued	Boolean	
ProductID	BigInt	Y
Picture	Image	

The *LeftList* box is loaded initially with the *ProductName* field of the *Products* table.

1. Associate next code with the *When loading* form event: after initial loading the *LeftList* control will be dissociated from its bound field. Additionally 3 constants are defined at module level.

```

Const cstForm           = "Products_FastSelect"
Const cstLeftList       = "LeftList"
Const cstRightList      = "RightList"

```

```

Sub InitFirstList(poEvent As Object)
'Activated when form opened
Dim ocList As Object, sSource As String

    Set ocList = Forms(cstForm).Controls(cstLeftList)
    sSource = Join(ocList.ItemData, ";")
    ocList.RowSourceType = com.sun.star.form.ListSourceType.VALUELIST
    ocList.RowSource = sSource
End Sub

```

2. Associate the *Execute action* event of each of the 4 buttons with next macro:

```

Sub MoveItems(poEvent As Object)
Dim oEvent As Object, ocLeftList As Object, ocRightList As Object
Dim i As Integer, vLeft() As Variant, vRight() As Variant, vSelected() As Variant
Dim iMaxDim As Integer, iLeftSize As Integer, iRightSize As Integer, sSource As String

    Set oEvent = Events(poEvent)
    Set ocLeftList = Forms(cstForm).Controls(cstLeftList)
    Set ocRightList = Forms(cstForm).Controls(cstRightList)
    ' Initial load of arrays
    vLeft = ocLeftList.ItemData
    vRight = ocRightList.ItemData
    iLeftSize = UBound(vLeft)
    iRightSize = UBound(vRight)
    iMaxDim = iLeftSize + 1 + iRightSize + 1
    ReDim Preserve vLeft(iMaxDim)
    ReDim Preserve vRight(iMaxDim)

    Select Case oEvent.Source.Name
        Case "MoveOneRight"
            vSelected = ocLeftList.Selected
            For i = 0 To iLeftSize
                If vSelected(i) Then
                    iRightSize = iRightSize + 1
                    vRight(iRightSize) = vLeft(i)
                    vLeft(i) = ""
                End If
            Next i
        Case "MoveAllRight"
            For i = 0 To iLeftSize
                iRightSize = iRightSize + 1
                vRight(iRightSize) = vLeft(i)
                vLeft(i) = ""
            Next i
        Case "MoveOneLeft"
            vSelected = ocRightList.Selected
            For i = 0 To iRightSize
                If vSelected(i) Then
                    iLeftSize = iLeftSize + 1
                    vLeft(iLeftSize) = vRight(i)
                    vRight(i) = ""
                End If
            Next i
        Case "MoveAllLeft"
            For i = 0 To iRightSize
                iLeftSize = iLeftSize + 1
                vLeft(iLeftSize) = vRight(i)
                vRight(i) = ""
            Next i
    End Select
End Sub

```

```

End Select

'Reload listboxes
sSource = ""
For i = 0 To iLeftSize
    If vLeft(i) <> "" Then sSource = sSource & vLeft(i) & ";"
Next i
If Len(sSource) = 0 Then ocLeftList.RowSource = "" Else ocLeftList.RowSource = Left(sSource, Len(sSource) - 1)
sSource = ""
For i = 0 To iRightSize
    If vRight(i) <> "" Then sSource = sSource & vRight(i) & ";"
Next i
If Len(sSource) = 0 Then ocRightList.RowSource = "" Else ocRightList.RowSource = Left(sSource, Len(sSource) - 1)

'Deselect listboxes
For i = 0 To ocLeftList.ListCount - 1
    setSelected(ocLeftList, False, i)
Next i
For i = 0 To ocRightList.ListCount - 1
    setSelected(ocRightList, False, i)
Next i
End Sub

```

## See also

**ListBox**

**ListCount**

**RowSource**

**RowSourceType**

**Selected**

**Event**

## Refer to ...

Basic module	Form	Form event	Control	Control event	Comments
FastSelect	Products_FastSelect	When loading	MoveOneRight MoveAllRight MoveOneLeft MoveAllLeft	Execute action	

Bookmark this page » » [MoveItemsBetweenListboxes](#)